

# Selektionsbeschränkungen

---

## Übersicht

- ◆ verschiedene Arten von Verben
- ◆ der Begriff der Valenz
- ◆ Selektionsrestriktionen
  - ◆ einfach
  - ◆ hierarchisch
- ◆ Implementation mit Prolog-DCGs

## Zweck

- ◆ Verstehen der verwendeten Technik
- ◆ in der Lage sein, eine DCG um Selektionsbeschränkungen zu erweitern

# Valenz

---

In den trivialen Beispiel-Grammatiken für das Parsing haben wir Verben nicht näher unterschieden.

Daher waren folgende Konstruktionen allesamt zulässig:

- ◆ \*Die Katze regnet.
- ◆ \*Die Katze gibt.
- ◆ \*Die Katze singt einen Hund.
- ◆ \*Die Katze träumt dem Hund einen Knochen.



\* Mit einem Stern werden in der Linguistik ungrammatische Sätze/Konstruktionen markiert.

# Valenz

Eine einfache Lösung wäre, die Verben nach ihrer *Valenz* (Wertigkeit) zu unterscheiden.

- ◆ nullwertige (avalente) Verben: *regnen, schneien, donnern*
  - ◆ Es regnet. *»es« kann nicht für »das Kätzchen« stehen, sondern ist sog. Expletivum*
  - ◆ \*Die Katze regnet.
  - ◆ \*Die Katze regnet einen Hund.
  - ◆ \*Die Katze regnet einem Hund, dass sie ihn mag.
- ◆ einwertige (monovalente) Verben: *schlafen, träumen*
  - ◆ \*Es schläft.
  - ◆ Die Katze schläft.
  - ◆ \*Die Katze schläft einen Hund.
  - ◆ \*Die Katze schläft einem Hund, dass sie ihn mag.

# Valenz

---

**Eine einfache Lösung wäre, die Verben nach ihrer Valenz (Wertigkeit) zu unterscheiden.**

- ◆ zweiwertige (bivalente/transitive) Verben: *lieben, verlassen*
  - ◆ \*Es liebt.
  - ◆ \*Die Katze liebt.
  - ◆ Die Katze liebt einen Hund.
  - ◆ \*Die Katze liebt einem Hund, dass sie ihn mag.
- ◆ dreiwertige (trivalente) Verben: *mitteilen, schenken, bezeichnen*
  - ◆ \*Es teilt mit.
  - ◆ ? Die Katze teilt mit.
  - ◆ \*Die Katze teilt einen/einem Hund mit.
  - ◆ Die Katze teilt einem Hund mit, dass sie ihn mag.

# Valenz

---

## Der Begriff der Valenz

- ◆ aus der Chemie übernommen
- ◆ Valenz rein als Anzahl der Argumentstellen ist veraltet (1959), neuere Ansätze sind differenzierter
- ◆ »die Fähigkeit eines Lexems (z.B. eines Verbs, Adjektivs, Substantivs), seine syntaktischen Umgebungen vorzustrukturieren, indem es anderen Konstituenten im Satz Bedingungen bezüglich ihrer grammatischen Eigenschaften auferlegt«
  - ◆ aus dem Eintrag zu »Valenz«, in: Hadumod Bußmann [Hrsg.]: Lexikon der Sprachwissenschaft. Stuttgart: Kröner, 1990. ISBN 3-520-45202-2.

# Selektionsrestriktionen

---

**Die Anzahl der Argumente ist nicht das einzige Kriterium, ob ein Satz akzeptabel ist.**

**Verben treffen weitere sog. *Selektionsrestriktionen*:**

- ◆ nur Belebtes kann verdursten:
  - ◆ Die Katze verdurstet.
  - ◆ Die Kokospalme verdurstet.
  - ◆ \*Die Schreibtischlampe verdurstet.
  - ◆ \*Die Linsensuppe verdurstet.
- ◆ hingegen kann Belebtes und Unbelebtes existieren:
  - ◆ Die Katze existiert.
  - ◆ Die Linsensuppe existiert.

# Selektionsrestriktionen

## Implementation mit einer Prolog-DCG:

```
s --> np(Belebtheit), vp(Belebtheit).  
np(Belebtheit) --> det, n(Belebtheit).  
vp(Belebtheit) --> v(Belebtheit).
```

```
det --> [die].
```

```
n(belebt) --> [katze].  
n(belebt) --> [kokospalme].  
n(unbelebt) --> [schreibtischlampe].  
n(unbelebt) --> [linsensuppe].
```

```
v(belebt) --> [verdurstet].  
v(_) --> [existiert].
```

```
test :-  
    phrase(s, Satz),  
    write(Satz), nl,  
    fail.  
test.
```

```
?- test.
```

```
[die,katze,verdurstet]  
[die,katze,existiert]  
[die,kokospalme,verdurstet]  
[die,kokospalme,existiert]  
[die,schreibtischlampe,existiert]  
[die,linsensuppe,existiert]
```

# Hierarchische Selektionsrestriktionen




## Weitere Selektionsbeschränkungen:

- ◆ Tiere können quieken, Pflanzen und Gegenstände nicht:
  - ◆ Die Katze quiekt.
  - ◆ \*Die Kokospalme quiekt.
  - ◆ \*Die Schreibtischlampe quiekt.
  - ◆ \*Die Linsensuppe quiekt.
- ◆ Tiere und Pflanzen können verdursten, nicht aber Gegenstände:
  - ◆ Die Katze verdurstet.
  - ◆ Die Kokospalme verdurstet.
  - ◆ \*Die Schreibtischlampe verdurstet.
  - ◆ \*Die Linsensuppe verdurstet.



# Hierarchische Selektionsrestriktionen

Es kann hilfreich sein, diese Beschränkungen hierarchisch zu definieren.




# Hierarchische Selektionsrestriktionen

## Selektionsbeschränkungen mit Einfacher Vererbung lassen sich in Prolog mit offenen Listen implementieren.

- ◆ Einfachvererbung (*simple inheritance*): Ein Knoten erbt von max. einem Elternteil → baum-förmige Struktur, wie in diesem Beispiel
- ◆ Mehrfachvererbung (*multiple inheritance*): Ein Knoten erbt Eigenschaften von mehreren Elternteilen — ist komplizierter zu implementieren

`v([entitaet,gegenstand,fluessig|_]) --> [gefriert].`

`n([entitaet,gegenstand,fluessig,suppe]) --> [linsensuppe].`



# Hierarchische Selektionsrestriktionen

## Implementation mit einer Prolog-DCG:


```
s --> np(Sel), vp(Sel).  
np(Sel) --> det, n(Sel).  
vp(Sel) --> v(Sel).
```

```
det --> [die].
```

```
n([entitaet, lebewesen, tier, katze]) --> [katze].  
n([entitaet, lebewesen, pflanze, palme]) --> [kokospalme].  
n([entitaet, gegenstand, fest]) --> [schreibtischlampe].  
n([entitaet, gegenstand, fluessig, suppe]) --> [linsensuppe].
```

```
v([entitaet, lebewesen|_]) --> [verdurstet].  
v([entitaet|_]) --> [existiert].  
v([entitaet, lebewesen, tier|_]) --> [quiekt].  
v([entitaet, gegenstand, fluessig|_]) --> [gefriert].
```


```
test :-  
    phrase(s, Satz),  
    write(Satz),  
    nl,  
    fail.  
test.
```



?- test.

```
[die, katze, verdurstet]  
[die, katze, existiert]  
[die, katze, quiekt]  
[die, kokospalme, verdurstet]  
[die, kokospalme, existiert]  
[die, schreibtischlampe, existiert]  
[die, linsensuppe, existiert]  
[die, linsensuppe, gefriert]
```

# Hierarchische Selektionsrestriktionen



Vereinfachter Auszug aus: H. Schumacher [Hrsg.]: Verben in Feldern. Valenzwörterbuch zur Syntax und Semantik deutscher Verben. Berlin: De Gruyter, 1986.

# A: Selektionsrestriktionen

Programmiertechniken der Computerlinguistik 2 · Sommersemester 1999

## 1. Nomen: Heute geschlossen

Bei genauem Betrachten der Folien fällt auf, dass die Selektionsrestriktionen von Verben mit offenen, jene von Nomina dagegen mit geschlossenen Listen deklariert wurden — zum Beispiel:

```
v([entitaet, gegenstand, fluessig | _]) --> [gefriert].
n([entitaet, gegenstand, fluessig, suppe]) --> [linsensuppe].
```

Warum?

## 2. Unsichtbare Gerüche (poetisch?)

Schreibe eine Definit-Klausel-Grammatik, bei der neben dem Subjekt auch das Objekt des Satzes durch Selektionsrestriktionen eingeschränkt wird. Stelle zudem sicher, dass an den passenden Stellen wie üblich Kongruenz (*Agreement*) verlangt wird. Die Grammatik enthalte sowohl transitive wie auch intransitive Verben.

Die folgenden Sätze sollen beispielsweise vom Startsymbol *S* ableitbar sein:

Die Frau sieht die Rose.  
Der Mann mag den Geruch.  
Die Frau träumt.  
Die Tulpe verblüht.

Hingegen sollen z.B. folgende Sätze nicht zur Sprache der Grammatik gehören:

\*Die Frau sieht den Geruch.  
\*Der Geruch mag den Mann.  
\*Die Tulpe verblüht den Mann.  
  
\*Die Frau sieht.  
\*Die Rose träumt.  
  
\*Der Frau mag die Mann.

Zeichne die Hierarchie auf, welche Du für die Selektionsrestriktionen verwendest. Die Grammatik lässt sich übrigens gut testen, indem Du Dir sämtliche ableitbaren Sätze anzeigen lässt.